In this Maple file, we compute the evolution equations for the Painlevé 3 equations using the compatibility equation of the Lax system. We also obtain the expression of the Lax matrices in the geometric gauge without apparent singularities.

Lax matrices in the oper gauge from previous Maple files

Summary of previous files: We have the expression for some coefficients of the Lax matrix L and of A.
The deformation operator is \mathcal{L}=\hbar (alphainf11\partial_{t_{\infty^{(1)},1} + alphainf21\partial_{t_{\infty^{(2)},1}+alpha011\partial_{t_{0^{(1)},1} +alpha021\partial_{t_{0^{(2)},1})))

```
> restart:
  with(LinearAlgebra):
  P042 := t011*t021;
  P032 := t010*t021+t011*t020;
  P012 := -1/2*(t010+t020)*(tinfty11+tinfty21)-1/2*(tinfty10-
  tinfty20)*(-tinfty21+tinfty11);
  P021 := t011+t021;
  P011 := t010+t020;
  Pinfty01 := -tinfty11-tinfty21;
  Pinfty02 := tinfty11*tinfty21;
  CoherenceEquation1 := tinfty10+tinfty20+t010+t020;

  P1:=lambda-> P021/lambda^2+P011/lambda+Pinfty01;
  P2:=lambda-> P042/lambda^4+P032/lambda^3+P022/lambda^2+
  P012/lambda+Pinfty02;
  dP1dlambda:=unapply(diff(P1(lambda),lambda),lambda):
  dP2dlambda:=unapply(diff(P2(lambda),lambda),lambda):

  tdP2:=unapply(P2(lambda)-P022/lambda^2-P012/lambda,lambda);

  L:=Matrix(2,2,0):
  L[1,1]:=0:
  L[1,2]:=1:
  L[2,1]:=-t011*t021/lambda^4  -(t010*t021+t020*t011)/lambda^3-
  H/lambda^2- tinfty11*tinfty21 - (tinfty11*tinfty20+tinfty21*
  tinfty10+h*tinfty11-h*p)/lambda- p*h/(lambda-q):
  L[2,2]:= (t011+t021)/lambda^2+ (t010+t020-2*h)/lambda -
  (tinfty11+tinfty21)   +h/(lambda-q):

  C01:=residue(L[2,1],lambda=0);
  C02:=residue(L[2,1]*lambda,lambda=0);

  A:=Matrix(2,2,0):
```

```
A[1,1]:=(alphainf11*tinfty21-alphainf21*tinfty11)/(tinfty11-
tinfty21)*lambda+ C+rho/(lambda-q)-(t011*alpha021-
t021*alpha011)/(t011-t021)/lambda:
A[1,2]:=(alphainf11-alphainf21)/(tinfty11-tinfty21)*lambda+nu+
mu/(lambda-q):
A[2,1]:=AA21(lambda):
A[2,2]:=AA22(lambda):
dAdlambda:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dAdlambda[i,j]:=diff
(A[i,j],lambda):  od: od:

L;
A;

nuMinus1:=factor(-residue(A[1,2]/lambda^2,lambda=infinity));
nu0:=factor(-residue(A[1,2]/lambda,lambda=infinity));
cinfty1:=factor(-residue(A[1,1]/lambda^2,lambda=infinity));
c01:=factor(residue(A[1,1],lambda=0));


Q2:=unapply(-p*(q-0)^2,lambda):
J:=Matrix(2,2,0):
J[1,1]:=1:
J[1,2]:=0:
J[2,1]:=Q2(lambda)/(lambda-q):
J[2,2]:=(lambda-0)^2/(lambda-q):
dJdlambda:=Matrix(2,2,0):
for i from 1 to 2 do for j from 1 to 2 do dJdlambda[i,j]:=diff
(J[i,j],lambda): od: od:
J:

LJ:=Matrix(2,2,0):
LJ[1,1]:=0:
LJ[1,2]:=0:
LJ[2,2]:=diff(J[2,2],q)*Lq+diff(J[2,2],p)*Lp+h*diff(J[2,2],t):
LJ[2,1]:=diff(J[2,1],q)*Lq+diff(J[2,1],p)*Lp+h*diff(J[2,1],t):
LJ:

checkL:=simplify(Multiply(Multiply(J,L),J^(-1))+h*Multiply
(dJdlambda,J^(-1))):
checkA:=simplify(Multiply(Multiply(J,A),J^(-1))+Multiply(LJ,J^
(-1))):
```

$$P042 := t011\, t021 \tag{1.1}$$
$$P032 := t010\, t021 + t011\, t020$$
$$P012 := -\frac{1}{2}\,(t010 + t020)\,(tinfty11 + tinfty21) - \frac{1}{2}\,(tinfty10 - tinfty20)\,(-tinfty21 + tinfty11)$$
$$P021 := t011 + t021$$
$$P011 := t010 + t020$$
$$Pinfty01 := -tinfty11 - tinfty21$$
$$Pinfty02 := tinfty11\, tinfty21$$
$$CoherenceEquation1 := tinfty10 + tinfty20 + t010 + t020$$
$$P1 := \lambda \rightarrow \frac{P021}{\lambda^2} + \frac{P011}{\lambda} + Pinfty01$$
$$P2 := \lambda \rightarrow \frac{P042}{\lambda^4} + \frac{P032}{\lambda^3} + \frac{P022}{\lambda^2} + \frac{P012}{\lambda} + Pinfty02$$
$$tdP2 := \lambda \rightarrow \frac{t011\, t021}{\lambda^4} + \frac{t010\, t021 + t011\, t020}{\lambda^3} + tinfty11\, tinfty21$$
$$C01 := -\frac{-h\, p\, q + h\, q\, tinfty11 + q\, tinfty10\, tinfty21 + q\, tinfty11\, tinfty20}{q}$$
$$C02 := -H$$
$$\left[\left[0,\ 1\right], \left[-\frac{t011\, t021}{\lambda^4} - \frac{t010\, t021 + t011\, t020}{\lambda^3} - \frac{H}{\lambda^2} - tinfty11\, tinfty21 - \frac{-h\, p + h\, tinfty11 + tinfty10\, tinfty21 + tinfty11\, tinfty20}{\lambda} - \frac{p\, h}{\lambda - q},\ \frac{t011 + t021}{\lambda^2} + \frac{t010 + t020 - 2\, h}{\lambda} - tinfty11 - tinfty21 + \frac{h}{\lambda - q}\right]\right]$$
$$\left[\left[\frac{(alphainf11\, tinfty21 - alphainf21\, tinfty11)\,\lambda}{-tinfty21 + tinfty11} + C + \frac{\rho}{\lambda - q} - \frac{t011\, \alpha021 - t021\, \alpha011}{(t011 - t021)\,\lambda},\ \frac{(alphainf11 - alphainf21)\,\lambda}{-tinfty21 + tinfty11} + \nu + \frac{\mu}{\lambda - q}\right], \left[AA21(\lambda),\ AA22(\lambda)\right]\right]$$
$$nuMinus1 := \frac{alphainf11 - alphainf21}{-tinfty21 + tinfty11}$$
$$\nu0 := \nu$$
$$cinfty1 := \frac{alphainf11\, tinfty21 - alphainf21\, tinfty11}{-tinfty21 + tinfty11}$$
$$c01 := -\frac{t011\, \alpha021 - t021\, \alpha011}{t011 - t021}$$

Solving the compatibility equations to obtain the Hamiltonian evolutions.

The compatibility equation is \mathcal{L}L=h\partial_\lambda A+[A,L]
Since the first line of L is trivial, we may easily obtain A[2,1] et A[2,2] to obtain the full expression for A

```
> LL:=h*dAdlambda+(Multiply(A,L)-Multiply(L,A)):
  Entry11:=LL[1,1]:
  Entry12:=LL[1,2]:

  AA21:=unapply(solve(Entry11=0,AA21(lambda)),lambda):
  AA21bis:=h*dAdlambda[1,1]+A[1,2]*L[2,1]:

  simplify(AA21(lambda)-AA21bis);
  AA22:=unapply(solve(Entry12=0,AA22(lambda)),lambda):
  AA22bis:=h*dAdlambda[1,2]+A[1,1]+A[1,2]*L[2,2]:

  simplify(AA22(lambda)-AA22bis);
  simplify(Entry11);
  simplify(Entry12);
  LL:=h*dAdlambda+(Multiply(A,L)-Multiply(L,A)):
```

$$\begin{array}{c}0\\0\\0\\0\end{array} \quad (2.1)$$

We now compute the action of \mathcal{L} on L[2,2] et L[2,1] to obtain the evolution equations

Evolution of entry L_{2,2}

```
> Entry22:=simplify(LL[2,2]):
  Entry22TermLambdaMinusqCube:=factor(residue(Entry22*(lambda-q)
  ^2,lambda=q));
  Entry22TermLambdaMinusqSquare:=factor(residue(Entry22*(lambda-
  q),lambda=q));
  Entry22TermLambdaMinusq:=factor(residue(Entry22,lambda=q));
  Entry22TermLambdaZeroMinus4:=factor(residue(Entry22*lambda^3,
  lambda=0));
  Entry22TermLambdaZeroMinus3:=factor(residue(Entry22*lambda^2,
  lambda=0));
  Entry22TermLambdaZeroMinus2:=factor(residue(Entry22*lambda,
  lambda=0));
  Entry22TermLambdaZeroMinus1:=factor(residue(Entry22,lambda=0));
  Entry22TermLambdaInfty2:=factor(-residue(Entry22/lambda^3,
  lambda=infinity));
  Entry22TermLambdaInfty1:=factor(-residue(Entry22/lambda^2,
  lambda=infinity));
```

```
Entry22TermLambdaInfty0:=factor(-residue(Entry22/lambda,lambda=
infinity));

simplify( Entry22-(Entry22TermLambdaMinusqSquare/(lambda-q)^2+
Entry22TermLambdaMinusq/(lambda-q)
+Entry22TermLambdaZeroMinus4/lambda^4+
Entry22TermLambdaZeroMinus3/lambda^3+
Entry22TermLambdaZeroMinus2/lambda^2+
Entry22TermLambdaZeroMinus1/lambda
+Entry22TermLambdaInfty0+Entry22TermLambdaInfty1*lambda+
Entry22TermLambdaInfty2*lambda^2) );
L[2,2];
```

$$Entry22TermLambdaMinusqCube := 0 \tag{2.2}$$

$$Entry22TermLambdaMinusqSquare := \frac{1}{q^2\,(-tinfty21 + tinfty11)}\Big(\big(-h\,\nu\,q^2\,tinfty11 + h\,\nu\,q^2\,tinfty21 - h\,q^3\,alphainf11 + h\,q^3\,alphainf21 + \mu\,q^2\,tinfty11^2 - \mu\,q^2\,tinfty21^2 + 2\,h\,\mu\,q\,tinfty11 - 2\,h\,\mu\,q\,tinfty21 - \mu\,q\,t010\,tinfty11 + \mu\,q\,t010\,tinfty21 - \mu\,q\,t020\,tinfty11 + \mu\,q\,t020\,tinfty21 - 2\,q^2\,\rho\,tinfty11 + 2\,q^2\,\rho\,tinfty21 - \mu\,t011\,tinfty11 + \mu\,t011\,tinfty21 - \mu\,t021\,tinfty11 + \mu\,t021\,tinfty21\big)\,h\Big)$$

$$Entry22TermLambdaMinusq := 0$$

$$Entry22TermLambdaZeroMinus4 := 0$$

$$Entry22TermLambdaZeroMinus3 := \frac{2\,(t011 + t021)\,(-\nu\,q + \mu)\,h}{q}$$

$$Entry22TermLambdaZeroMinus2 := -\frac{1}{q^2\,(-tinfty21 + tinfty11)\,(t011 - t021)}\Big(h\,\big(-2\,h\,\nu\,q^2\,t011\,tinfty11 + 2\,h\,\nu\,q^2\,t011\,tinfty21 + 2\,h\,\nu\,q^2\,t021\,tinfty11 - 2\,h\,\nu\,q^2\,t021\,tinfty21 + \nu\,q^2\,t010\,t011\,tinfty11 - \nu\,q^2\,t010\,t011\,tinfty21 - \nu\,q^2\,t010\,t021\,tinfty11 + \nu\,q^2\,t010\,t021\,tinfty21 + \nu\,q^2\,t011\,t020\,tinfty11 - \nu\,q^2\,t011\,t020\,tinfty21 - \nu\,q^2\,t020\,t021\,tinfty11 + \nu\,q^2\,t020\,t021\,tinfty21 + 2\,h\,\mu\,q\,t011\,tinfty11 - 2\,h\,\mu\,q\,t011\,tinfty21 - 2\,h\,\mu\,q\,t021\,tinfty11 + 2\,h\,\mu\,q\,t021\,tinfty21 - \mu\,q\,t010\,t011\,tinfty11 + \mu\,q\,t010\,t011\,tinfty21 + \mu\,q\,t010\,t021\,tinfty11 - \mu\,q\,t010\,t021\,tinfty21 - \mu\,q\,t011\,t020\,tinfty11 + \mu\,q\,t011\,t020\,tinfty21 + \mu\,q\,t020\,t021\,tinfty11 - \mu\,q\,t020\,t021\,tinfty21 + q^2\,t011^2\,alphainf11 - q^2\,t011^2\,alphainf21 - 2\,q^2\,t011\,\alpha 021\,tinfty11 + 2\,q^2\,t011\,\alpha 021\,tinfty21 - q^2\,t021^2\,alphainf11 + q^2\,t021^2\,alphainf21 + 2\,q^2\,t021\,\alpha 011\,tinfty11 - 2\,q^2\,t021\,\alpha 011\,tinfty21 - \mu\,t011^2\,tinfty11 + \mu\,t011^2\,tinfty21 + \mu\,t021^2\,tinfty11 - \mu\,t021^2\,tinfty21\big)\Big)$$

$$Entry22TermLambdaZeroMinus1 := 0$$

$$Entry22TermLambdaInfty2 := 0$$

$$Entry22TermLambdaInfty1 := 0$$

$$Entry22TermLambdaInfty0 := -h\,(alphainf11 + alphainf21)$$

$$0$$

$$\frac{t011 + t021}{\lambda^2} + \frac{t010 + t020 - 2\,h}{\lambda} - tinfty11 - tinfty21 + \frac{h}{\lambda - q}$$

Since the deformation operator is mathcal{L}=\hbar (alphainf11\partial_{t_{\infty^{(1)},1} + alphainf21\partial_{t_{\infty^{(2)},1}+alpha011\partial_{t_{0^{(1)},1} +alpha021\partial_{t_{0^{(2)},1}))), the double pole at lambda=0 is h*(alpha011+alpha021)

```
> solve({Entry22TermLambdaZeroMinus3,Entry22TermLambdaZeroMinus2-h*(alpha011+alpha021)},{nu,mu});
mu := q^2*(t011*alphainf11-t011*alphainf21-t021*alphainf11+t021*alphainf21+alpha011*tinfty11-alpha011*tinfty21-alpha021*tinfty11+alpha021*tinfty21)/(t011*tinfty11-t011*tinfty21-t021*tinfty11+t021*tinfty21);

nu := q*(t011*alphainf11-t011*alphainf21-t021*alphainf11+t021*alphainf21+alpha011*tinfty11-alpha011*tinfty21-alpha021*tinfty11+alpha021*tinfty21)/(t011*tinfty11-t011*tinfty21-t021*tinfty11+t021*tinfty21);
simplify(Entry22TermLambdaZeroMinus3);
simplify(Entry22TermLambdaZeroMinus2-h*(alpha011+alpha021));
```

$$\Big\{\mu = \Big(\big(t011\,alphainf11 - t011\,alphainf21 - t021\,alphainf11 + t021\,alphainf21 + \alpha011\,tinfty11 - \alpha011\,tinfty21 - \alpha021\,tinfty11 + \alpha021\,tinfty21\big)\,q^2\Big) \Big/ (t011\,tinfty11 - t011\,tinfty21 - t021\,tinfty11 + t021\,tinfty21),\ \nu = \Big(\big(t011\,alphainf11 - t011\,alphainf21 - t021\,alphainf11 + t021\,alphainf21 + \alpha011\,tinfty11 - \alpha011\,tinfty21 - \alpha021\,tinfty11 + \alpha021\,tinfty21\big)\,q\Big) \Big/ (t011\,tinfty11 - t011\,tinfty21 - t021\,tinfty11 + t021\,tinfty21)\Big\} \quad \textbf{(2.3)}$$

$$\mu := \Big(\big(t011\,alphainf11 - t011\,alphainf21 - t021\,alphainf11 + t021\,alphainf21 + \alpha011\,tinfty11 - \alpha011\,tinfty21 - \alpha021\,tinfty11 + \alpha021\,tinfty21\big)\,q^2\Big) \Big/ (t011\,tinfty11 - t011\,tinfty21 - t021\,tinfty11 + t021\,tinfty21)$$

$$\nu := \Big(\big(t011\,alphainf11 - t011\,alphainf21 - t021\,alphainf11 + t021\,alphainf21 + \alpha011\,tinfty11 - \alpha011\,tinfty21 - \alpha021\,tinfty11 + \alpha021\,tinfty21\big)\,q\Big) \Big/ (t011\,tinfty11 - t011\,tinfty21 - t021\,tinfty11 + t021\,tinfty21)$$

$$0$$

$$0$$

```
> Lq:=factor(Entry22TermLambdaMinusqSquare/h):
Lqbis:=-2*rho-((alphainf11-alphainf21)/(tinfty11-tinfty21)+(alpha011-alpha021)/(t011-t021))*q^2*P1(q)+(alpha011-alpha021)*q/(t011-t021)*h;
factor(simplify(series(Lq-Lqbis,q=0)));
```

(2.4)

$$Lqbis := -2\,\rho - \left(\frac{alphainf11 - alphainf21}{-tinfty21 + tinfty11} + \frac{\alpha 011 - \alpha 021}{t011 - t021}\right) q^2 \left(\frac{t011 + t021}{q^2} + \frac{t010 + t020}{q} - tinfty11 - tinfty21\right) + \frac{(\alpha 011 - \alpha 021)\, q\, h}{t011 - t021} \tag{2.4}$$

$$0$$

Evolution of \mathcal{L}[L[2,1]]

```
> Entry21:=simplify(LL[2,1]):
Entry21TermLambdaMinusqCube:=factor(residue(Entry21*(lambda-q)
^2,lambda=q));
Entry21TermLambdaMinusqSquare:=factor(residue(Entry21*(lambda-
q),lambda=q));
Entry21TermLambdaMinusq:=factor(residue(Entry21,lambda=q));
Entry21TermLambdaZeroMinus5:=factor(residue(Entry21*lambda^4,
lambda=0));
Entry21TermLambdaZeroMinus4:=factor(residue(Entry21*lambda^3,
lambda=0));
Entry21TermLambdaZeroMinus3:=factor(residue(Entry21*lambda^2,
lambda=0));
Entry21TermLambdaZeroMinus2:=factor(residue(Entry21*lambda,
lambda=0));
Entry21TermLambdaZeroMinus1:=factor(residue(Entry21,lambda=0));
Entry21TermLambdaInfty3:=factor(-residue(Entry21/lambda^4,
lambda=infinity));
Entry21TermLambdaInfty2:=factor(-residue(Entry21/lambda^3,
lambda=infinity));
Entry21TermLambdaInfty1:=factor(-residue(Entry21/lambda^2,
lambda=infinity));
Entry21TermLambdaInfty0:=factor(-residue(Entry21/lambda,lambda=
infinity));

simplify( Entry21-(Entry21TermLambdaMinusqCube/(lambda-q)^3+
Entry21TermLambdaMinusqSquare/(lambda-q)^2+
Entry21TermLambdaMinusq/(lambda-q)
+Entry21TermLambdaZeroMinus5/lambda^5+
Entry21TermLambdaZeroMinus4/lambda^4+
Entry21TermLambdaZeroMinus3/lambda^3+
Entry21TermLambdaZeroMinus2/lambda^2+
Entry21TermLambdaZeroMinus1/lambda
+Entry21TermLambdaInfty0+Entry21TermLambdaInfty1*lambda+
Entry21TermLambdaInfty2*lambda^2+Entry21TermLambdaInfty3*
lambda^3) );
L[2,1];
```

$$
\begin{aligned}
&\mathit{Entry21TermLambdaMinusqCube} := \frac{1}{(-\mathit{tinfty21}+\mathit{tinfty11})\,(\mathit{t011}-\mathit{t021})}\Big(3\,h^{2}\,\big(p\,q^{2} \\
&\quad \mathit{t011}\,\mathit{alphainf11} - p\,q^{2}\,\mathit{t011}\,\mathit{alphainf21} - p\,q^{2}\,\mathit{t021}\,\mathit{alphainf11} + p\,q^{2}\,\mathit{t021}\,\mathit{alphainf21} \\
&\quad + p\,q^{2}\,\alpha\mathit{011}\,\mathit{tinfty11} - p\,q^{2}\,\alpha\mathit{011}\,\mathit{tinfty21} - p\,q^{2}\,\alpha\mathit{021}\,\mathit{tinfty11} + p\,q^{2}\,\alpha\mathit{021}\,\mathit{tinfty21} \\
&\quad + \rho\,\mathit{t011}\,\mathit{tinfty11} - \rho\,\mathit{t011}\,\mathit{tinfty21} - \rho\,\mathit{t021}\,\mathit{tinfty11} + \rho\,\mathit{t021}\,\mathit{tinfty21}\big)\Big)
\end{aligned}
\tag{2.5}
$$

$$
\begin{aligned}
&\mathit{Entry21TermLambdaMinusqSquare} := \frac{1}{q^{2}\,(-\mathit{tinfty21}+\mathit{tinfty11})\,(\mathit{t011}-\mathit{t021})}\Big(\big(2\,q^{4} \\
&\quad \mathit{t011}\,\mathit{alphainf11}\,\mathit{tinfty11}\,\mathit{tinfty21} - 2\,q^{4}\,\mathit{t011}\,\mathit{alphainf21}\,\mathit{tinfty11}\,\mathit{tinfty21} \\
&\quad - 2\,q^{4}\,\mathit{t021}\,\mathit{alphainf11}\,\mathit{tinfty11}\,\mathit{tinfty21} + 2\,q^{4}\,\mathit{t021}\,\mathit{alphainf21}\,\mathit{tinfty11}\,\mathit{tinfty21} \\
&\quad + 2\,q^{4}\,\alpha\mathit{011}\,\mathit{tinfty11}^{2}\,\mathit{tinfty21} - 2\,q^{4}\,\alpha\mathit{011}\,\mathit{tinfty11}\,\mathit{tinfty21}^{2} \\
&\quad - 2\,q^{4}\,\alpha\mathit{021}\,\mathit{tinfty11}^{2}\,\mathit{tinfty21} + 2\,q^{4}\,\alpha\mathit{021}\,\mathit{tinfty11}\,\mathit{tinfty21}^{2} - h\,p\,q^{3}\,\alpha\mathit{011}\,\mathit{tinfty11} \\
&\quad + h\,p\,q^{3}\,\alpha\mathit{011}\,\mathit{tinfty21} + h\,p\,q^{3}\,\alpha\mathit{021}\,\mathit{tinfty11} - h\,p\,q^{3}\,\alpha\mathit{021}\,\mathit{tinfty21} \\
&\quad + 2\,h\,q^{3}\,\mathit{t011}\,\mathit{alphainf11}\,\mathit{tinfty11} - 2\,h\,q^{3}\,\mathit{t011}\,\mathit{alphainf21}\,\mathit{tinfty11} \\
&\quad - 2\,h\,q^{3}\,\mathit{t021}\,\mathit{alphainf11}\,\mathit{tinfty11} + 2\,h\,q^{3}\,\mathit{t021}\,\mathit{alphainf21}\,\mathit{tinfty11} \\
&\quad + 2\,h\,q^{3}\,\alpha\mathit{011}\,\mathit{tinfty11}^{2} - 2\,h\,q^{3}\,\alpha\mathit{011}\,\mathit{tinfty11}\,\mathit{tinfty21} - 2\,h\,q^{3}\,\alpha\mathit{021}\,\mathit{tinfty11}^{2} \\
&\quad + 2\,h\,q^{3}\,\alpha\mathit{021}\,\mathit{tinfty11}\,\mathit{tinfty21} + 2\,q^{3}\,\mathit{t011}\,\mathit{alphainf11}\,\mathit{tinfty10}\,\mathit{tinfty21} \\
&\quad + 2\,q^{3}\,\mathit{t011}\,\mathit{alphainf11}\,\mathit{tinfty11}\,\mathit{tinfty20} - 2\,q^{3}\,\mathit{t011}\,\mathit{alphainf21}\,\mathit{tinfty10}\,\mathit{tinfty21} \\
&\quad - 2\,q^{3}\,\mathit{t011}\,\mathit{alphainf21}\,\mathit{tinfty11}\,\mathit{tinfty20} - 2\,q^{3}\,\mathit{t021}\,\mathit{alphainf11}\,\mathit{tinfty10}\,\mathit{tinfty21} \\
&\quad - 2\,q^{3}\,\mathit{t021}\,\mathit{alphainf11}\,\mathit{tinfty11}\,\mathit{tinfty20} + 2\,q^{3}\,\mathit{t021}\,\mathit{alphainf21}\,\mathit{tinfty10}\,\mathit{tinfty21} \\
&\quad + 2\,q^{3}\,\mathit{t021}\,\mathit{alphainf21}\,\mathit{tinfty11}\,\mathit{tinfty20} + 2\,q^{3}\,\alpha\mathit{011}\,\mathit{tinfty10}\,\mathit{tinfty11}\,\mathit{tinfty21} \\
&\quad - 2\,q^{3}\,\alpha\mathit{011}\,\mathit{tinfty10}\,\mathit{tinfty21}^{2} + 2\,q^{3}\,\alpha\mathit{011}\,\mathit{tinfty11}^{2}\,\mathit{tinfty20} \\
&\quad - 2\,q^{3}\,\alpha\mathit{011}\,\mathit{tinfty11}\,\mathit{tinfty20}\,\mathit{tinfty21} - 2\,q^{3}\,\alpha\mathit{021}\,\mathit{tinfty10}\,\mathit{tinfty11}\,\mathit{tinfty21} \\
&\quad + 2\,q^{3}\,\alpha\mathit{021}\,\mathit{tinfty10}\,\mathit{tinfty21}^{2} - 2\,q^{3}\,\alpha\mathit{021}\,\mathit{tinfty11}^{2}\,\mathit{tinfty20} \\
&\quad + 2\,q^{3}\,\alpha\mathit{021}\,\mathit{tinfty11}\,\mathit{tinfty20}\,\mathit{tinfty21} - q^{2}\,\rho\,\mathit{t011}\,\mathit{tinfty11}^{2} + q^{2}\,\rho\,\mathit{t011}\,\mathit{tinfty21}^{2} \\
&\quad + q^{2}\,\rho\,\mathit{t021}\,\mathit{tinfty11}^{2} - q^{2}\,\rho\,\mathit{t021}\,\mathit{tinfty21}^{2} + 2\,H\,q^{2}\,\mathit{t011}\,\mathit{alphainf11} \\
&\quad - 2\,H\,q^{2}\,\mathit{t011}\,\mathit{alphainf21} - 2\,H\,q^{2}\,\mathit{t021}\,\mathit{alphainf11} + 2\,H\,q^{2}\,\mathit{t021}\,\mathit{alphainf21} \\
&\quad + 2\,H\,q^{2}\,\alpha\mathit{011}\,\mathit{tinfty11} - 2\,H\,q^{2}\,\alpha\mathit{011}\,\mathit{tinfty21} - 2\,H\,q^{2}\,\alpha\mathit{021}\,\mathit{tinfty11} \\
&\quad + 2\,H\,q^{2}\,\alpha\mathit{021}\,\mathit{tinfty21} - 2\,h\,q\,\rho\,\mathit{t011}\,\mathit{tinfty11} + 2\,h\,q\,\rho\,\mathit{t011}\,\mathit{tinfty21} \\
&\quad + 2\,h\,q\,\rho\,\mathit{t021}\,\mathit{tinfty11} - 2\,h\,q\,\rho\,\mathit{t021}\,\mathit{tinfty21} + q\,\rho\,\mathit{t010}\,\mathit{t011}\,\mathit{tinfty11} \\
&\quad - q\,\rho\,\mathit{t010}\,\mathit{t011}\,\mathit{tinfty21} - q\,\rho\,\mathit{t010}\,\mathit{t021}\,\mathit{tinfty11} + q\,\rho\,\mathit{t010}\,\mathit{t021}\,\mathit{tinfty21} \\
&\quad + q\,\rho\,\mathit{t011}\,\mathit{t020}\,\mathit{tinfty11} - q\,\rho\,\mathit{t011}\,\mathit{t020}\,\mathit{tinfty21} - q\,\rho\,\mathit{t020}\,\mathit{t021}\,\mathit{tinfty11} \\
&\quad + q\,\rho\,\mathit{t020}\,\mathit{t021}\,\mathit{tinfty21} + 2\,q\,\mathit{t010}\,\mathit{t011}\,\mathit{t021}\,\mathit{alphainf11} \\
&\quad - 2\,q\,\mathit{t010}\,\mathit{t011}\,\mathit{t021}\,\mathit{alphainf21} - 2\,q\,\mathit{t010}\,\mathit{t021}^{2}\,\mathit{alphainf11} \\
&\quad + 2\,q\,\mathit{t010}\,\mathit{t021}^{2}\,\mathit{alphainf21} + 2\,q\,\mathit{t010}\,\mathit{t021}\,\alpha\mathit{011}\,\mathit{tinfty11} - 2\,q\,\mathit{t010}\,\mathit{t021}\,\alpha\mathit{011}\,\mathit{tinfty21} \\
&\quad - 2\,q\,\mathit{t010}\,\mathit{t021}\,\alpha\mathit{021}\,\mathit{tinfty11} + 2\,q\,\mathit{t010}\,\mathit{t021}\,\alpha\mathit{021}\,\mathit{tinfty21} + 2\,q\,\mathit{t011}^{2}\,\mathit{t020}\,\mathit{alphainf11} \\
&\quad - 2\,q\,\mathit{t011}^{2}\,\mathit{t020}\,\mathit{alphainf21} - 2\,q\,\mathit{t011}\,\mathit{t020}\,\mathit{t021}\,\mathit{alphainf11} \\
&\quad + 2\,q\,\mathit{t011}\,\mathit{t020}\,\mathit{t021}\,\mathit{alphainf21} + 2\,q\,\mathit{t011}\,\mathit{t020}\,\alpha\mathit{011}\,\mathit{tinfty11} \\
&\quad - 2\,q\,\mathit{t011}\,\mathit{t020}\,\alpha\mathit{011}\,\mathit{tinfty21} - 2\,q\,\mathit{t011}\,\mathit{t020}\,\alpha\mathit{021}\,\mathit{tinfty11} \\
&\quad + 2\,q\,\mathit{t011}\,\mathit{t020}\,\alpha\mathit{021}\,\mathit{tinfty21} + \rho\,\mathit{t011}^{2}\,\mathit{tinfty11} - \rho\,\mathit{t011}^{2}\,\mathit{tinfty21} - \rho\,\mathit{t021}^{2}\,\mathit{tinfty11} \\
&\quad + \rho\,\mathit{t021}^{2}\,\mathit{tinfty21} + 2\,\mathit{t011}^{2}\,\mathit{t021}\,\mathit{alphainf11} - 2\,\mathit{t011}^{2}\,\mathit{t021}\,\mathit{alphainf21}
\end{aligned}
$$

$$- 2\, t011\, t021^{2}\, alphainf11 + 2\, t011\, t021^{2}\, alphainf21 + 2\, t011\, t021\, \alpha011\, tinfty11$$
$$- 2\, t011\, t021\, \alpha011\, tinfty21 - 2\, t011\, t021\, \alpha021\, tinfty11 + 2\, t011\, t021\, \alpha021\, tinfty21\Big)$$
$$h\Big)$$

$$Entry21TermLambdaMinusq := -\frac{1}{q^{3}\, (-tinfty21 + tinfty11)\, (t011 - t021)}\Big(\Big($$
$$-h\, p\, q^{3}\, \alpha011\, tinfty11 + h\, p\, q^{3}\, \alpha011\, tinfty21 + h\, p\, q^{3}\, \alpha021\, tinfty11$$
$$- h\, p\, q^{3}\, \alpha021\, tinfty21 + h\, q^{3}\, t011\, alphainf11\, tinfty11 + h\, q^{3}\, t011\, alphainf11\, tinfty21$$
$$- 2\, h\, q^{3}\, t011\, alphainf21\, tinfty11 - h\, q^{3}\, t021\, alphainf11\, tinfty11$$
$$- h\, q^{3}\, t021\, alphainf11\, tinfty21 + 2\, h\, q^{3}\, t021\, alphainf21\, tinfty11 + h\, q^{3}\, \alpha011\, tinfty11^{2}$$
$$- h\, q^{3}\, \alpha011\, tinfty11\, tinfty21 - h\, q^{3}\, \alpha021\, tinfty11^{2} + h\, q^{3}\, \alpha021\, tinfty11\, tinfty21$$
$$+ q^{3}\, t011\, alphainf11\, tinfty10\, tinfty21 + q^{3}\, t011\, alphainf11\, tinfty11\, tinfty20$$
$$- q^{3}\, t011\, alphainf21\, tinfty10\, tinfty21 - q^{3}\, t011\, alphainf21\, tinfty11\, tinfty20$$
$$- q^{3}\, t021\, alphainf11\, tinfty10\, tinfty21 - q^{3}\, t021\, alphainf11\, tinfty11\, tinfty20$$
$$+ q^{3}\, t021\, alphainf21\, tinfty10\, tinfty21 + q^{3}\, t021\, alphainf21\, tinfty11\, tinfty20$$
$$+ q^{3}\, \alpha011\, tinfty10\, tinfty11\, tinfty21 - q^{3}\, \alpha011\, tinfty10\, tinfty21^{2}$$
$$+ q^{3}\, \alpha011\, tinfty11^{2}\, tinfty20 - q^{3}\, \alpha011\, tinfty11\, tinfty20\, tinfty21$$
$$- q^{3}\, \alpha021\, tinfty10\, tinfty11\, tinfty21 + q^{3}\, \alpha021\, tinfty10\, tinfty21^{2}$$
$$- q^{3}\, \alpha021\, tinfty11^{2}\, tinfty20 + q^{3}\, \alpha021\, tinfty11\, tinfty20\, tinfty21 + 2\, H\, q^{2}\, t011\, alphainf11$$
$$- 2\, H\, q^{2}\, t011\, alphainf21 - 2\, H\, q^{2}\, t021\, alphainf11 + 2\, H\, q^{2}\, t021\, alphainf21$$
$$+ 2\, H\, q^{2}\, \alpha011\, tinfty11 - 2\, H\, q^{2}\, \alpha011\, tinfty21 - 2\, H\, q^{2}\, \alpha021\, tinfty11$$
$$+ 2\, H\, q^{2}\, \alpha021\, tinfty21 - 2\, h\, q\, \rho\, t011\, tinfty11 + 2\, h\, q\, \rho\, t011\, tinfty21$$
$$+ 2\, h\, q\, \rho\, t021\, tinfty11 - 2\, h\, q\, \rho\, t021\, tinfty21 + h\, q\, t011\, \alpha021\, tinfty11$$
$$- h\, q\, t011\, \alpha021\, tinfty21 - h\, q\, t021\, \alpha011\, tinfty11 + h\, q\, t021\, \alpha011\, tinfty21$$
$$+ q\, \rho\, t010\, t011\, tinfty11 - q\, \rho\, t010\, t011\, tinfty21 - q\, \rho\, t010\, t021\, tinfty11$$
$$+ q\, \rho\, t010\, t021\, tinfty21 + q\, \rho\, t011\, t020\, tinfty11 - q\, \rho\, t011\, t020\, tinfty21$$
$$- q\, \rho\, t020\, t021\, tinfty11 + q\, \rho\, t020\, t021\, tinfty21 + 3\, q\, t010\, t011\, t021\, alphainf11$$
$$- 3\, q\, t010\, t011\, t021\, alphainf21 - 3\, q\, t010\, t021^{2}\, alphainf11$$
$$+ 3\, q\, t010\, t021^{2}\, alphainf21 + 3\, q\, t010\, t021\, \alpha011\, tinfty11 - 3\, q\, t010\, t021\, \alpha011\, tinfty21$$
$$- 3\, q\, t010\, t021\, \alpha021\, tinfty11 + 3\, q\, t010\, t021\, \alpha021\, tinfty21 + 3\, q\, t011^{2}\, t020\, alphainf11$$
$$- 3\, q\, t011^{2}\, t020\, alphainf21 - 3\, q\, t011\, t020\, t021\, alphainf11$$
$$+ 3\, q\, t011\, t020\, t021\, alphainf21 + 3\, q\, t011\, t020\, \alpha011\, tinfty11$$
$$- 3\, q\, t011\, t020\, \alpha011\, tinfty21 - 3\, q\, t011\, t020\, \alpha021\, tinfty11$$
$$+ 3\, q\, t011\, t020\, \alpha021\, tinfty21 + 2\, \rho\, t011^{2}\, tinfty11 - 2\, \rho\, t011^{2}\, tinfty21$$
$$- 2\, \rho\, t021^{2}\, tinfty11 + 2\, \rho\, t021^{2}\, tinfty21 + 4\, t011^{2}\, t021\, alphainf11$$
$$- 4\, t011^{2}\, t021\, alphainf21 - 4\, t011\, t021^{2}\, alphainf11 + 4\, t011\, t021^{2}\, alphainf21$$
$$+ 4\, t011\, t021\, \alpha011\, tinfty11 - 4\, t011\, t021\, \alpha011\, tinfty21 - 4\, t011\, t021\, \alpha021\, tinfty11$$
$$+ 4\, t011\, t021\, \alpha021\, tinfty21\Big)\, h\Big)$$

$$Entry21TermLambdaZeroMinus5 := 0$$
$$Entry21TermLambdaZeroMinus4 := -\left(t011\, \alpha021 + t021\, \alpha011\right) h$$
$$Entry21TermLambdaZeroMinus3 := -\left(t010\, \alpha021 + t020\, \alpha011\right) h$$

$$Entry21TermLambdaZeroMinus2 := \frac{1}{q^2\ (-tinfty21 + tinfty11)\ (t011 - t021)} (h\ ($$
$$-q^2\ t011^2\ alphainf11\ tinfty21 + q^2\ t011^2\ alphainf21\ tinfty11 + q^2\ t011\ \alpha021\ tinfty11^2$$
$$- q^2\ t011\ \alpha021\ tinfty21^2 + q^2\ t021^2\ alphainf11\ tinfty21 - q^2\ t021^2\ alphainf21\ tinfty11$$
$$- q^2\ t021\ \alpha011\ tinfty11^2 + q^2\ t021\ \alpha011\ tinfty21^2 + h\ q\ t011\ \alpha021\ tinfty11$$
$$- h\ q\ t011\ \alpha021\ tinfty21 - h\ q\ t021\ \alpha011\ tinfty11 + h\ q\ t021\ \alpha011\ tinfty21$$
$$+ q\ t010\ t011\ t021\ alphainf11 - q\ t010\ t011\ t021\ alphainf21 - q\ t010\ t021^2\ alphainf11$$
$$+ q\ t010\ t021^2\ alphainf21 + q\ t010\ t021\ \alpha011\ tinfty11 - q\ t010\ t021\ \alpha011\ tinfty21$$
$$- q\ t010\ t021\ \alpha021\ tinfty11 + q\ t010\ t021\ \alpha021\ tinfty21 + q\ t011^2\ t020\ alphainf11$$
$$- q\ t011^2\ t020\ alphainf21 - q\ t011\ t020\ t021\ alphainf11 + q\ t011\ t020\ t021\ alphainf21$$
$$+ q\ t011\ t020\ \alpha011\ tinfty11 - q\ t011\ t020\ \alpha011\ tinfty21 - q\ t011\ t020\ \alpha021\ tinfty11$$
$$+ q\ t011\ t020\ \alpha021\ tinfty21 + \rho\ t011^2\ tinfty11 - \rho\ t011^2\ tinfty21 - \rho\ t021^2\ tinfty11$$
$$+ \rho\ t021^2\ tinfty21 + 2\ t011^2\ t021\ alphainf11 - 2\ t011^2\ t021\ alphainf21$$
$$- 2\ t011\ t021^2\ alphainf11 + 2\ t011\ t021^2\ alphainf21 + 2\ t011\ t021\ \alpha011\ tinfty11$$
$$- 2\ t011\ t021\ \alpha011\ tinfty21 - 2\ t011\ t021\ \alpha021\ tinfty11 + 2\ t011\ t021\ \alpha021\ tinfty21))$$

$$Entry21TermLambdaZeroMinus1 := \frac{1}{q^3\ (-tinfty21 + tinfty11)\ (t011 - t021)} (($$
$$-h\ p\ q^3\ \alpha011\ tinfty11 + h\ p\ q^3\ \alpha011\ tinfty21 + h\ p\ q^3\ \alpha021\ tinfty11$$
$$- h\ p\ q^3\ \alpha021\ tinfty21 + 2\ h\ q^3\ t011\ alphainf11\ tinfty21$$
$$- 2\ h\ q^3\ t011\ alphainf21\ tinfty11 - 2\ h\ q^3\ t021\ alphainf11\ tinfty21$$
$$+ 2\ h\ q^3\ t021\ alphainf21\ tinfty11 + h\ q^3\ \alpha011\ tinfty11^2 - h\ q^3\ \alpha011\ tinfty11\ tinfty21$$
$$- h\ q^3\ \alpha021\ tinfty11^2 + h\ q^3\ \alpha021\ tinfty11\ tinfty21 - q^3\ t010\ t011\ alphainf11\ tinfty21$$
$$+ q^3\ t010\ t011\ alphainf21\ tinfty11 + q^3\ t010\ t021\ alphainf11\ tinfty21$$
$$- q^3\ t010\ t021\ alphainf21\ tinfty11 - q^3\ t011\ t020\ alphainf11\ tinfty21$$
$$+ q^3\ t011\ t020\ alphainf21\ tinfty11 + q^3\ t020\ t021\ alphainf11\ tinfty21$$
$$- q^3\ t020\ t021\ alphainf21\ tinfty11 + q^3\ \alpha011\ tinfty10\ tinfty11\ tinfty21$$
$$- q^3\ \alpha011\ tinfty10\ tinfty21^2 + q^3\ \alpha011\ tinfty11^2\ tinfty20$$
$$- q^3\ \alpha011\ tinfty11\ tinfty20\ tinfty21 - q^3\ \alpha021\ tinfty10\ tinfty11\ tinfty21$$
$$+ q^3\ \alpha021\ tinfty10\ tinfty21^2 - q^3\ \alpha021\ tinfty11^2\ tinfty20$$
$$+ q^3\ \alpha021\ tinfty11\ tinfty20\ tinfty21 + 2\ H\ q^2\ t011\ alphainf11 - 2\ H\ q^2\ t011\ alphainf21$$
$$- 2\ H\ q^2\ t021\ alphainf11 + 2\ H\ q^2\ t021\ alphainf21 + 2\ H\ q^2\ \alpha011\ tinfty11$$
$$- 2\ H\ q^2\ \alpha011\ tinfty21 - 2\ H\ q^2\ \alpha021\ tinfty11 + 2\ H\ q^2\ \alpha021\ tinfty21$$
$$- 2\ h\ q\ \rho\ t011\ tinfty11 + 2\ h\ q\ \rho\ t011\ tinfty21 + 2\ h\ q\ \rho\ t021\ tinfty11$$
$$- 2\ h\ q\ \rho\ t021\ tinfty21 + h\ q\ t011\ \alpha021\ tinfty11 - h\ q\ t011\ \alpha021\ tinfty21$$
$$- h\ q\ t021\ \alpha011\ tinfty11 + h\ q\ t021\ \alpha011\ tinfty21 + q\ \rho\ t010\ t011\ tinfty11$$
$$- q\ \rho\ t010\ t011\ tinfty21 - q\ \rho\ t010\ t021\ tinfty11 + q\ \rho\ t010\ t021\ tinfty21$$
$$+ q\ \rho\ t011\ t020\ tinfty11 - q\ \rho\ t011\ t020\ tinfty21 - q\ \rho\ t020\ t021\ tinfty11$$
$$+ q\ \rho\ t020\ t021\ tinfty21 + 3\ q\ t010\ t011\ t021\ alphainf11$$
$$- 3\ q\ t010\ t011\ t021\ alphainf21 - 3\ q\ t010\ t021^2\ alphainf11$$
$$+ 3\ q\ t010\ t021^2\ alphainf21 + 3\ q\ t010\ t021\ \alpha011\ tinfty11 - 3\ q\ t010\ t021\ \alpha011\ tinfty21$$

$- 3\, q\, t010\, t021\, \alpha021\, tinfty11 + 3\, q\, t010\, t021\, \alpha021\, tinfty21 + 3\, q\, t011^2\, t020\, alphainf11$
$- 3\, q\, t011^2\, t020\, alphainf21 - 3\, q\, t011\, t020\, t021\, alphainf11$
$+ 3\, q\, t011\, t020\, t021\, alphainf21 + 3\, q\, t011\, t020\, \alpha011\, tinfty11$
$- 3\, q\, t011\, t020\, \alpha011\, tinfty21 - 3\, q\, t011\, t020\, \alpha021\, tinfty11$
$+ 3\, q\, t011\, t020\, \alpha021\, tinfty21 + 2\, \rho\, t011^2\, tinfty11 - 2\, \rho\, t011^2\, tinfty21$
$- 2\, \rho\, t021^2\, tinfty11 + 2\, \rho\, t021^2\, tinfty21 + 4\, t011^2\, t021\, alphainf11$
$- 4\, t011^2\, t021\, alphainf21 - 4\, t011\, t021^2\, alphainf11 + 4\, t011\, t021^2\, alphainf21$
$+ 4\, t011\, t021\, \alpha011\, tinfty11 - 4\, t011\, t021\, \alpha011\, tinfty21 - 4\, t011\, t021\, \alpha021\, tinfty11$
$+ 4\, t011\, t021\, \alpha021\, tinfty21 \big)\, h \big)$

$$Entry21TermLambdaInfty3 := 0$$
$$Entry21TermLambdaInfty2 := 0$$
$$Entry21TermLambdaInfty1 := 0$$
$$Entry21TermLambdaInfty0 := -(alphainf11\, tinfty21 + alphainf21\, tinfty11)\, h$$
$$0$$
$$-\frac{t011\, t021}{\lambda^4} - \frac{t010\, t021 + t011\, t020}{\lambda^3} - \frac{H}{\lambda^2} - tinfty11\, tinfty21$$
$$- \frac{-h\, p + h\, tinfty11 + tinfty10\, tinfty21 + tinfty11\, tinfty20}{\lambda} - \frac{p\, h}{\lambda - q}$$

```
> rho:=factor(solve(Entry21TermLambdaMinusqCube,rho));
  simplify(rho+p*q*nu);
  simplify(Entry21TermLambdaMinusqCube);
  LH:=simplify(-Entry21TermLambdaZeroMinus1):
  EquationPoleSimple:=simplify(-h*(alphainf11*tinfty20+
  alphainf21*tinfty10+h*alphainf11-Lp)-
  Entry21TermLambdaZeroMinus1):
```

$$\rho := -\frac{1}{(-tinfty21 + tinfty11)\,(t011 - t021)} \big(p\, q^2\, \big(t011\, alphainf11 - t011\, alphainf21 - t021\, alphainf11 + t021\, alphainf21 + \alpha011\, tinfty11 - \alpha011\, tinfty21 - \alpha021\, tinfty11 + \alpha021\, tinfty21\big)\big) \quad \textbf{(2.6)}$$
$$0$$
$$0$$

```
> LpFunction:=unapply(-Entry21TermLambdaMinusq/h,H):
> Equation1:=simplify(Entry21TermLambdaMinusqSquare-(-p*h*Lq)):
> Hsol:=solve(Equation1,H):
  Hsolbis:=-q^2*p^2+q^2*P1(q)*p-q^2*(P2(q)-P022/q^2)-p*q*h-
  tinfty11*q*h;
  factor(series(Hsol-Hsolbis+(tinfty11+tinfty21)
  /2*CoherenceEquation1*q,q=0));
```

$$Hsolbis := -q^2\, p^2 + q^2 \left(\frac{t011 + t021}{q^2} + \frac{t010 + t020}{q} - tinfty11 - tinfty21 \right) p \quad \textbf{(2.7)}$$

$$-q^2\left(\frac{t011\,t021}{q^4}+\frac{t010\,t021+t011\,t020}{q^3}\right.$$
$$+\frac{1}{q}\left(-\frac{1}{2}\,(t010+t020)\,(tinfty11+tinfty21)-\frac{1}{2}\,(tinfty10-tinfty20)\,(\right.$$
$$\left.\left.-tinfty21+tinfty11)\right)+tinfty11\,tinfty21\right)-h\,p\,q-tinfty11\,q\,h$$

$$0$$

```
> Lp:=factor(simplify(LpFunction(Hsol))):
Lpbis:=
((alphainf11-alphainf21)/(tinfty11-tinfty21)+(alpha011-
alpha021)/(t011-t021) )*(-2*q*p^2+diff(q^2*P1(q),q)*p-diff(q^2*
P2(q),q))
-alphainf11*h- (alpha011-alpha021)*h*p/(t011-t021)-h*(alpha011-
alpha021)*tinfty11/(t011-t021)-(t021*alpha011-t011*alpha021)*
h/q^2/(t011-t021):

Lpter:=((alphainf11-alphainf21)/(tinfty11-tinfty21)+(alpha011-
alpha021)/(t011-t021) )*(
-2*q*p^2+((t010+t020)-2*(tinfty11+tinfty21)*q)*p+2*t011*
t021/q^3+(t010*t021+t011*t020)/q^2 +1/2*((t010+t020)*(tinfty11+
tinfty21)+(tinfty10-tinfty20)*(tinfty11-tinfty21))-2*tinfty11*
tinfty21*q)
-alphainf11*h- (alpha011-alpha021)*h*p/(t011-t021)-h*(alpha011-
alpha021)*tinfty11/(t011-t021)-(t021*alpha011-t011*alpha021)*
h/q^2/(t011-t021):

factor(series(Lp-Lpbis+1/2*(tinfty11+tinfty21)*(-alphainf11+
alphainf21)/(tinfty21-tinfty11)*CoherenceEquation1-1/2*
(tinfty11+tinfty21)*(alpha011-alpha021)/(t021-t011)*
CoherenceEquation1,p=0));
factor(Lp-Lpter-1/2*(tinfty11+tinfty21)*(alphainf11-alphainf21)
/(tinfty21-tinfty11)*CoherenceEquation1-1/2*(tinfty11+tinfty21)
*(alpha011-alpha021)/(t021-t011)*CoherenceEquation1);
```

$$0 \tag{2.8}$$
$$0$$

```
> factor(Lqbis-  (   ((alphainf11-alphainf21)/(tinfty11-tinfty21)
+(alpha011-alpha021)/(t011-t021)  )*(2*q^2*p-q^2*P1(q))   +
(alpha011-alpha021)*q*h/(t011-t021))   );
```

$$0 \tag{2.9}$$

We get that

L[q]= ((alphainf11-alphainf21)/(tinfty11-tinfty21)+(alpha011-alpha021)/(t011-t021))*(2*q^2*

p-q^2*P1(q)) -(alpha011-alpha021)*q*h/(t011-t021)
L[p] = ((alphainf11-alphainf21)/(tinfty11-tinfty21)+(alpha011-alpha021)/(t011-t021))*(-2*q*p^2+diff(q^2*P1(q),q)*p-diff(q^2*P2(q),q)) -alphainf11*h - (alpha011-alpha021)*h*p/(t011-t021)-h*(alpha011-alpha021)*tinfty11/(t011-t021)-(t021*alpha011-t011*alpha021)*h/q^2/(t011-t021)

```
> Hamiltonian:=  ((alphainf11-alphainf21)/(tinfty11-tinfty21)+
  (alpha011-alpha021)/(t011-t021))*(q^2*p^2-q^2*P1(q)*p+q^2*P2(q)
  )+alphainf11*q*h+(alpha011-alpha021)*p*q*h/(t011-t021)+h*
  (alpha011-alpha021)*tinfty11/(t011-t021)*q-(t021*alpha011-t011*
  alpha021)*h/q/(t011-t021);
  factor(simplify(diff(Hamiltonian,p)-Lq));
  factor(simplify(diff(Hamiltonian,q)+Lp+(tinfty11+tinfty21)*(-
  alphainf11+alphainf21)/(tinfty21-tinfty11)
  /2*CoherenceEquation1)
  -1/2*(tinfty11+tinfty21)*(alpha011-alpha021)/(t021-t011)*
  CoherenceEquation1);

  Hamiltonianbis:= mu*(p^2-P1(q)*p+h*p*(2/q) +tdP2(q) )-h*nu0*p-
  h*nuMinus1*q*p -h*c01/q- h*cinfty1*q
  +nu0*(tinfty11*tinfty20+tinfty21*tinfty10+h*tinfty11) :

  factor(simplify(Lp-(-diff(Hamiltonianbis,q))));
  simplify(Lq-(diff(Hamiltonianbis,p)));
```

$$Hamiltonian := \left(\frac{alphainf11 - alphainf21}{-tinfty21 + tinfty11} + \frac{\alpha 011 - \alpha 021}{t011 - t021}\right)\Bigg(q^2 p^2 - q^2\left(\frac{t011 + t021}{q^2} + \frac{t010 + t020}{q} - tinfty11 - tinfty21\right) p + q^2\Bigg(\frac{t011\, t021}{q^4} + \frac{t010\, t021 + t011\, t020}{q^3} + \frac{P022}{q^2} + \frac{1}{q}\Bigg(-\frac{1}{2}(t010 + t020)(tinfty11 + tinfty21) - \frac{1}{2}(tinfty10 - tinfty20)(-tinfty21 + tinfty11)\Bigg) + tinfty11\, tinfty21\Bigg)\Bigg) + alphainf11\, q\, h + \frac{(\alpha 011 - \alpha 021)\, p\, q\, h}{t011 - t021} + \frac{h\,(\alpha 011 - \alpha 021)\, tinfty11\, q}{t011 - t021} - \frac{(-t011\, \alpha 021 + t021\, \alpha 011)\, h}{q\,(t011 - t021)} \quad (2.10)$$

$$0$$
$$0$$
$$0$$
$$0$$

Decomposition of the tangent space

```
> tdp:=p-P1(q)/2:
```

```
Ltdp:=Lp-1/2*diff(P1(q),q)*Lq-1/2*(h*alphainf11*diff(P1(q),
tinfty11)+h*alphainf21*diff(P1(q),tinfty21)+h*alpha011*diff(P1
(q),t011)+h*alpha021*diff(P1(q),t021)):
Ltdpbis:=-((alphainf11-alphainf21)/(tinfty11-tinfty21)+
(alpha011-alpha021)/(t011-t021))*(2*q*tdp^2+diff(q^2*(P2(q)-P1
(q)^2/4),q)
+h/2*(tinfty11-tinfty21))-(alpha011-alpha021)*h*tdp/(t011-t021)
:
factor(series(factor(simplify(Ltdp-Ltdpbis+(tinfty11+tinfty21)*
(alphainf11-alphainf21)/2/(tinfty11-tinfty21)
*CoherenceEquation1)
+(tinfty11+tinfty21)*(alpha011-alpha021)/2/(t011-t021)*
CoherenceEquation1),h=0));
```

$$0 \tag{3.1}$$

```
> Lpfunction:=unapply(simplify(Lp),
alphainf11,alphainf21,alpha011,alpha021):
Ltdpfunction:=unapply(simplify(Ltdp),alphainf11,alphainf21,
alpha011,alpha021):
Lqfunction:=unapply(simplify(Lq),alphainf11,alphainf21,
alpha011,alpha021):
cinfty1function:=unapply(cinfty1,alphainf11,alphainf21,
alpha011,alpha021):
c01function:=unapply(c01,alphainf11,alphainf21,alpha011,
alpha021):
nuMinus1function:=unapply(nuMinus1,alphainf11,alphainf21,
alpha011,alpha021):
```

```
> factor(Ltdpfunction(1,1,0,0));
factor(Lqfunction(1,1,0,0));
factor(cinfty1function(1,1,0,0));
factor(c01function(1,1,0,0));
factor(nuMinus1function(1,1,0,0));
```

$$0 \tag{3.2}$$
$$0$$
$$-1$$
$$0$$
$$0$$

```
> factor(Ltdpfunction(0,0,1,1));
factor(Lqfunction(0,0,1,1));
factor(cinfty1function(0,0,1,1));
factor(c01function(0,0,1,1));
factor(nuMinus1function(0,0,1,1));
```

$$0 \tag{3.3}$$
$$0$$
$$0$$

$$-1$$
$$0$$

Expression of the Lax matrices in the geometric gauge after the symplectic reduction and the Painlevé 3 equation

Expression of the geometric Lax matrix in the gauge without apparent singularities after symplectic reduction

```
> tinfty21:=-tinfty11:
  tinfty20:=-tinfty10:
  t021:=-t011:
  t020:=-t010:
  tinfty11:=1:
  t011:=t/2:
  H:=Hsol:
  C:=0:
  q:=checkq:
  p:=checkp:
  dcheckqdt:=Lq/h:
  dcheckpdt:=Lp/h:

  alphainf11:=0:
  alphainf21:=0:
  alpha011:=1/2:
  alpha021:=-1/2:
```

```
> G1:=Matrix(2,2,0):
  G1[1,1]:=1:
  G1[2,2]:=1:
  G1[1,2]:=0:
  G1[2,1]:=g1*lambda+g0:
  g1:=tinfty11;
  g0:=checkq+tinfty10;

  dG1dlambda:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dG1dlambda[i,j]:=diff
  (G1[i,j],lambda): od: od:

  dG1dtau:=Matrix(2,2,0):
  for i from 1 to 2 do for j from 1 to 2 do dG1dtau[i,j]:=diff(G1
  [i,j],t)+diff(G1[i,j],checkq)*dcheckqdt+diff(G1[i,j],checkp)
  *dcheckpdt : od: od:
```

```
tdL:=simplify(Multiply(Multiply(G1,checkL),G1^(-1))+h*Multiply
(dG1dlambda,G1^(-1))):
tdA:=simplify(Multiply(Multiply(G1,checkA),G1^(-1))+h*Multiply
(dG1dtau,G1^(-1))):

checkL:=simplify(checkL);
checkA:=simplify(checkA);
tdL:=simplify(tdL);
tdA:=simplify(tdA);
```

$$g1 := 1 \tag{4.1}$$

$$g0 := checkq + tinfty10$$

$$\left[\left[\frac{checkp\ checkq^2}{\lambda^2}, \frac{\lambda - checkq}{\lambda^2}\right], \left[\frac{1}{4}\frac{1}{\lambda^2\ checkq^2}\left(4\ checkp^2\ checkq^5 + 4\ checkp^2\ checkq^4\ \lambda + 4\ \lambda^2\ checkq^3 - 4\ \lambda^2\ (h - \lambda - 2\ tinfty10)\ checkq^2 + (-4\ \lambda\ t\ t010 - t^2)\ checkq - \lambda\ t^2\right), -\frac{checkp\ checkq^2}{\lambda^2}\right]\right]$$

$$\left[\left[-\frac{checkp\ checkq^2}{\lambda\ t}, \frac{checkq}{t\ \lambda}\right], \left[\frac{1}{4}\frac{-4\ checkp^2\ checkq^4 + 8\ \lambda\ checkq^3 - 4\ \lambda\ (h - \lambda - 2\ tinfty10)\ checkq^2 + t^2}{\lambda\ checkq\ t}, \frac{checkp\ checkq^2}{\lambda\ t}\right]\right]$$

$$\left[\left[\frac{(checkp + 1)\ checkq^2 + tinfty10\ checkq - \lambda\ (\lambda + tinfty10)}{\lambda^2}, \frac{\lambda - checkq}{\lambda^2}\right], \left[\frac{1}{4}\frac{1}{\lambda^2\ checkq^2}\left(4\ (checkp + 1)^2\ checkq^5 + 4\ ((checkp + 1)\ \lambda + 2\ tinfty10)\ (checkp + 1)\ checkq^4 + 4\ tinfty10^2\ checkq^3 - 4\ \lambda\ tinfty10^2\ checkq^2 + (-4\ \lambda\ t\ t010 - t^2)\ checkq - \lambda\ t^2\right), \frac{(-checkp - 1)\ checkq^2 - tinfty10\ checkq + \lambda\ (\lambda + tinfty10)}{\lambda^2}\right]\right]$$

$$\left[\left[-\frac{(checkp\ checkq + checkq + \lambda + tinfty10)\ checkq}{\lambda\ t}, \frac{checkq}{t\ \lambda}\right],\right.$$

$$\left[\frac{1}{4}\,\frac{1}{\lambda\,checkq\,t}\Big(\big((2\,checkp+2)\,checkq^2+2\,tinfty10\,checkq+t\big)\,\big((-2\,checkp-2)\,checkq^2-2\,tinfty10\,checkq+t\big)\Big),\right.$$

$$\left.\frac{\big(checkp\,checkq+checkq+\lambda+tinfty10\big)\,checkq}{\lambda\,t}\right]\Bigg]$$

```
> series(tdL[1,1],lambda=infinity);
  series(tdL[1,2],lambda=infinity);
  residue(tdL[2,1]/lambda,lambda=infinity);
```

$$-1-\frac{tinfty10}{\lambda}+\frac{(checkp+1)\,checkq^2+tinfty10\,checkq}{\lambda^2} \qquad \textbf{(4.2)}$$

$$\frac{1}{\lambda}-\frac{checkq}{\lambda^2}$$

$$0$$

Reduced Hamiltonian evolutions

```
> simplify(series(dcheckqdt,checkp=0));
  simplify(series(dcheckpdt,checkp=0));
```

$$\frac{checkq}{t}+\frac{2\,checkq^2}{t\,h}\,checkp \qquad \textbf{(4.3)}$$

$$\frac{1}{2}\,\frac{4\,checkq^4+(-2\,h+4\,tinfty10)\,checkq^3-2\,checkq\,t010\,t-t^2}{checkq^3\,t\,h}-\frac{1}{t}\,checkp-\frac{2\,checkq}{t\,h}\,checkp^2$$